

TP n°2

Les exercices que vous devez rendre sont à la fin de l'énoncé, vous pouvez les traiter au fur et à mesure du TP (les outils pour l'exercice 1 sont dans la première partie, pour l'exercice 2, dans les deux premières parties et pour les exercices 3, 4 et 5, dans la troisième partie) ou à la fin du TP.

1 Tracés en deux dimensions

Tracé de courbes

Dans le TP n°1, nous avons vu que tracer la courbe représentative de la fonction tangente sur $[-5, 5]$ créait une fenêtre aux dimensions inadaptées. Nous allons voir comment corriger ce problème.

Dans la console SCILAB

```
-- > x=linspace(-5,5,200);
-- > y=tan(x);
-- > plot(x,y)                                // On peut changer les propriétés de la
                                                // courbe grâce au menu "Édition" ou
                                                // avec les commandes suivantes.
-- > a=get('current_axes');                    // On peut aussi taper a=gca();
-- > a.data_bounds                             // Affiche [xmin,ymin;xmax,ymax].
-- > a.data_bounds= [-%pi,-3;%pi,3];          // Quelles sont les bornes de tracé?
```

En fait, toutes les propriétés de la fenêtre sont accessibles via `get('current_axes')`. Par exemple, on peut donner un titre à la courbe avec

Dans la console SCILAB

```
-- > a.title.text="Mon_titre";
```

Le résultat n'est pas forcément satisfaisant, le tracé fait apparaître des droites verticales en $-\frac{\pi}{2}$ et $\frac{\pi}{2}$. Pour les faire disparaître, on va éviter des intervalles autour de $-\frac{\pi}{2}$ et $\frac{\pi}{2}$.

Dans la console SCILAB

```
-- > clf
-- > x1 = linspace(-5, -%pi/2-0.2, 100);
-- > x2 = linspace(-%pi/2+0.2, %pi/2-0.2, 100);
-- > x3 = linspace(%pi/2+0.2, 5, 100);
-- > plot(x1, tan(x1))                          // Revient à faire y1=tan(x1);
                                                // plot(x1,y1)
-- > plot(x2, tan(x2))
-- > plot(x3, tan(x3), 'red')                    // Choix de la couleur de la courbe.
-- > a=get('current_axes');
-- > a.data_bounds= [-%pi,-3;%pi,3];
```

Tracé de figures

La fonction `plot` de SCILAB prend en argument deux vecteurs et relie les points de coordonnées correspondantes.

Dans la console SCILAB

```
-- > clf
-- > x = [0,1,2]
-- > y = [2,3,-2]
-- > plot(x,y) // Interpréter le tracé.
-- > clf
-- > x = [0,1,2,0]
-- > y = [2,3,-2,2]
-- > plot(x,y) // Quelle figure obtient-on?
```

2 Matrices

Dans la console SCILAB

```
-- > clf
-- > A = [1 2 ; -1 1] // On peut séparer avec des espaces
-- > B = [1,3,2 ; -1,0,0] // ou des virgules.
-- > A' // Transposition.
-- > A+A'
-- > A*B
-- > B*A // Pourquoi un message d'erreur?
-- > B'*A'
-- > A+1
-- > size(B)
-- > X=[0;1]
-- > A*X
-- > B'*X
-- > A^2 // Carré pour le produit matriciel.
-- > A.^2 // Opération coordonnée par coordonnée.
```

On peut utiliser le produit coordonnée par coordonnée pour tracer un polynôme.

Dans la console SCILAB

```
-- > clf
-- > x= linspace(-4,4,200);
-- > y= x.*x.*x +3*x -1; // On peut aussi écrire x.^3+3*x-1.
-- > plot(x,y)
```


Dans les versions plus anciennes de SCILAB, la commande `x^2` n'était pas applicable pour un vecteur ligne, désormais, elle revient à `x.^2` pour les *vecteurs lignes*.

3 Tracés en trois dimensions

Nous allons voir comment tracer des surfaces définies par des équations de la forme $z = f(x, y)$. Comme pour les fonctions d'une variable, SCILAB trace les surfaces grâce à un maillage.

Dans la console SCILAB

```
-- > clf
-- > X= linspace(-1,1,3)
-- > Y= linspace(-1,1,3)
-- > [x,y]=meshgrid(X,Y)
-- > z= x.* y; // Attention à ne pas oublier le point.
-- > surf(x,y,z) // Le maillage n'est pas assez fin.
```



Voici le maillage construit par la commande `meshgrid` de SCILAB à partir des vecteurs X et Y .

En ajoutant des points pour X et Y , on obtient un maillage et un tracé plus précis.

Dans la console SCILAB

```
-- > clf
-- > X= linspace(-1,1,20);
-- > Y= linspace(-1,1,20);
-- > [x,y]=meshgrid(X,Y)
-- > z= x.* y;
-- > surf(x,y,z) // Observer le point col
```

4 Exercices à rendre

Dans l'éditeur de textes SCINOTES (disponible dans le menu « Applications »), créez le fichier `TP2scilab.sci`. Vous devrez taper les instructions utiles à la résolution des exercices suivants dans ce fichier.

Au début de ce fichier, indiquez vos noms et prénoms en commentaire.

```
// Étudiant 1
```

```
// Étudiant 2
```

Indiquez le début de chaque exercice en commentaire.

```
// Exercice 1
```

Exercice 1

Tracez le carré de sommets $(0, 0)$, $(0, 1)$, $(1, 0)$ et $(1, 1)$.

Exercice 2

Sur le même graphique, tracer les courbes représentatives des fonctions $x \mapsto \sqrt{16 - x^2}$ et d'une autre couleur celle de $x \mapsto \frac{1}{x}$ pour $-4 \leq x \leq 4$. On supprimera les droites verticales éventuelles.

Exercice 3

Tracer les surfaces d'équations respectives $z = x^2 + y^2$ et $z = x^2 - y^2$ pour $-1 \leq x \leq 1$ et $-1 \leq y \leq 1$. Indiquez (en commentaire) les éventuels maxima locaux, minima locaux ou points cols.

Exercice 4

Tracer la surface d'équation $z = (x^2 - y^2)e^{-x^2 - y^2}$ pour $-2 \leq x \leq 2$ et $-2 \leq y \leq 2$. Combien cette surface admet-elle de minima locaux, de maxima locaux, de points cols ?

Exercice 5

Faire une représentation graphique de la surface d'équation $z = \sin(x) \cdot \sin(y)$ avec $0 \leq x \leq 4\pi$ et $0 \leq y \leq 4\pi$.

Enregistrez votre fichier, puis, grâce au navigateur web, envoyez-le par mail à l'adresse

`pierre.lezowski@math.u-bordeaux1.fr`.